In[30]:=
```
Clear["Global`*"]
```

Diagrams

In[ ]:=





In[ ]:=



Rotations

Length of a vector

In[134]:=
```
one = DiagonalMatrix [{1, 1}];
one  // MatrixForm
```

Out[135]//MatrixForm=

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

In[136]:=
```
vec1 = {vx, vy};
vec1  // MatrixForm
```

Out[137]//MatrixForm=

$$\begin{pmatrix} vx \\ vy \end{pmatrix}$$

In[138]:=
```
length1  = vec1.vec1
```

Out[138]= $vx^2 + vy^2$

In[139]:=
```
MatrixForm /@ {vec1, one, vec1}
```

Out[139]= $\left\{\begin{pmatrix} vx \\ vy \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} vx \\ vy \end{pmatrix}\right\}$

In[140]:=
```
length2 = vec1.one.vec1
```
Out[140]= $vx^2 + vy^2$

In[141]:=
```
length1 == length2
```
Out[141]= True

Length of a rotated vector

In[142]:=
```
m = {{Cos[θ], +Sin[θ]}, {-Sin[θ], Cos[θ]}};
m // MatrixForm
```
Out[143]//MatrixForm=

$$\begin{pmatrix} Cos[\theta] & Sin[\theta] \\ -Sin[\theta] & Cos[\theta] \end{pmatrix}$$

In[144]:=
```
vec2 = m.vec1;
vec2 // MatrixForm
```
Out[145]//MatrixForm=

$$\begin{pmatrix} vx\ Cos[\theta] + vy\ Sin[\theta] \\ vy\ Cos[\theta] - vx\ Sin[\theta] \end{pmatrix}$$

In[146]:=
```
MatrixForm /@ {vec2, one, vec2}
```
Out[146]=

$$\left\{\begin{pmatrix} vx\ Cos[\theta] + vy\ Sin[\theta] \\ vy\ Cos[\theta] - vx\ Sin[\theta] \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} vx\ Cos[\theta] + vy\ Sin[\theta] \\ vy\ Cos[\theta] - vx\ Sin[\theta] \end{pmatrix}\right\}$$

In[147]:=
```
length3 = vec2.one.vec2
```
Out[147]= $(vy\ Cos[\theta] - vx\ Sin[\theta])^2 + (vx\ Cos[\theta] + vy\ Sin[\theta])^2$

In[148]:=
```
length3 // Simplify
```
Out[148]= $vx^2 + vy^2$

Diagram Rotations

In[150]:=
```
xVec = {1, 0};
yVec = {0, 1};
m.xVec // MatrixForm
m.yVec // MatrixForm
```
Out[152]//MatrixForm=

$$\begin{pmatrix} Cos[\theta] \\ -Sin[\theta] \end{pmatrix}$$

Out[153]//MatrixForm=

$$\begin{pmatrix} Sin[\theta] \\ Cos[\theta] \end{pmatrix}$$

In[154]:=
```
makeVec[v_] := Arrow[{{0, 0}, v}]
```

In[155]:=
```
doPlot[θ0_] :=
 {Blue, makeVec[xVec], makeVec[yVec], Red, makeVec[m.xVec], makeVec[m.yVec]} /.
   {θ → θ0 Degree} // Graphics
```

In[160]:=
```
doPlot[-15]
```

Out[160]=



Boosts

Length of a vector

In[168]:=
```
g = DiagonalMatrix [{1, -1}];
g // MatrixForm
```

Out[169]//MatrixForm=

$$\begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}$$

In[170]:=
```
v1 = {vt, vx};
v1 // MatrixForm
```

Out[171]//MatrixForm=

$$\begin{pmatrix} vt \\ vx \end{pmatrix}$$

In[172]:=
```
MatrixForm /@ {v1, g, v1}
```

Out[172]=
$$\left\{\begin{pmatrix} vt \\ vx \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}, \begin{pmatrix} vt \\ vx \end{pmatrix}\right\}$$

In[173]:=
```
length4 = v1.g.v1
```

Out[173]=
$$vt^2 - vx^2$$

Length of a rotated vector

[Note, this is a symmetric matrix, NOT anti-symmetric.]

In[174]:=
```
b = {{Cosh[ψ], +Sinh[ψ]}, {+Sinh[ψ], Cosh[ψ]}};
b // MatrixForm
```

Out[175]//MatrixForm=

$$\begin{pmatrix} \text{Cosh}[\psi] & \text{Sinh}[\psi] \\ \text{Sinh}[\psi] & \text{Cosh}[\psi] \end{pmatrix}$$

In[176]:=
```
v2 = b.v1;
v2 // MatrixForm
```

Out[177]//MatrixForm=

$$\begin{pmatrix} \text{vt Cosh}[\psi] + \text{vx Sinh}[\psi] \\ \text{vx Cosh}[\psi] + \text{vt Sinh}[\psi] \end{pmatrix}$$

In[178]:=
```
MatrixForm /@ {v2, g, v2}
```

Out[178]=

$$\left\{\begin{pmatrix} \text{vt Cosh}[\psi] + \text{vx Sinh}[\psi] \\ \text{vx Cosh}[\psi] + \text{vt Sinh}[\psi] \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}, \begin{pmatrix} \text{vt Cosh}[\psi] + \text{vx Sinh}[\psi] \\ \text{vx Cosh}[\psi] + \text{vt Sinh}[\psi] \end{pmatrix}\right\}$$

In[181]:=
```
length5 = v2.g.v2 // Expand
```

Out[181]=

$$\text{vt}^2\ \text{Cosh}[\psi]^2 - \text{vx}^2\ \text{Cosh}[\psi]^2 - \text{vt}^2\ \text{Sinh}[\psi]^2 + \text{vx}^2\ \text{Sinh}[\psi]^2$$

In[182]:=
```
length5 // Simplify
```

Out[182]=

$$\text{vt}^2 - \text{vx}^2$$

Diagram Boosts

In[66]:=
```
b = {{Cosh[ψ], +Sinh[ψ]}, {+Sinh[ψ], Cosh[ψ]}};
b // MatrixForm
```

Out[67]//MatrixForm=

$$\begin{pmatrix} \text{Cosh}[\psi] & \text{Sinh}[\psi] \\ \text{Sinh}[\psi] & \text{Cosh}[\psi] \end{pmatrix}$$

In[68]:=
```
tVec1 = {1, 0};
xVec1 = {0, 1};
b.tVec1 // MatrixForm
b.xVec1 // MatrixForm
```

Out[70]//MatrixForm=

$$\begin{pmatrix} \text{Cosh}[\psi] \\ \text{Sinh}[\psi] \end{pmatrix}$$

Out[71]//MatrixForm=

$$\begin{pmatrix} \text{Sinh}[\psi] \\ \text{Cosh}[\psi] \end{pmatrix}$$

In[72]:=
```
doPlotBoost[ψ0_] :=
 {Blue, makeVec[tVec1], makeVec[xVec1], Red, makeVec[b.tVec1], makeVec[b.xVec1]} /.
   {ψ → ψ0} // Graphics
```

In[195]:=
```
doPlotBoost[0.4]
```

Out[195]=



Cosh & Sinh

In[74]:=
```
Plot[{Cosh[ψ], Sinh[ψ]}, {ψ, -2, 2}, PlotStyle → {Red, Blue}]
```

Out[74]=



In[75]:=
```
1 == Cosh[ψ]^2 - Sinh[ψ]^2 // Simplify
```

Out[75]=
```
True
```

In[197]:=
```
Plot[{Cosh[ψ]^2, Sinh[ψ]^2}, {ψ, -2, 2}, PlotStyle → {Red, Blue}]
```

Out[197]=



Boosts: $\gamma\ \beta$ representation

Length of a vector

In[77]:=
```
v1 = {vt, vx};
v1 // MatrixForm
```

Out[78]//MatrixForm=

$$\begin{pmatrix} vt \\ vx \end{pmatrix}$$

In[79]:=
```
MatrixForm /@ {v1, g, v1}
```

Out[79]= $\left\{\begin{pmatrix} vt \\ vx \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}, \begin{pmatrix} vt \\ vx \end{pmatrix}\right\}$

In[80]:=
```
length4 = v1.g.v1
```

Out[80]= $vt^2 - vx^2$

Length of a rotated vector

[Note, this is a symmetric matrix, NOT anti-symmetric.]

In[81]:=
```
b2 = {{γ, β γ}, {β γ, γ}};
b2 // MatrixForm
```

Out[82]//MatrixForm=

$$\begin{pmatrix} \gamma & \beta\gamma \\ \beta\gamma & \gamma \end{pmatrix}$$

In[83]:=
```
v2 = b2.v1;
v2 // MatrixForm
```

Out[84]//MatrixForm=

$$\begin{pmatrix} vt\,\gamma + vx\,\beta\,\gamma \\ vx\,\gamma + vt\,\beta\,\gamma \end{pmatrix}$$

In[85]:=
```
MatrixForm /@ {v2, g, v2}
```

Out[85]=

$$\left\{\begin{pmatrix} vt\,\gamma + vx\,\beta\,\gamma \\ vx\,\gamma + vt\,\beta\,\gamma \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}, \begin{pmatrix} vt\,\gamma + vx\,\beta\,\gamma \\ vx\,\gamma + vt\,\beta\,\gamma \end{pmatrix}\right\}$$

In[86]:=
```
length5 = v2.g.v2
```

Out[86]= $(-vx\,\gamma - vt\,\beta\,\gamma)\,(vx\,\gamma + vt\,\beta\,\gamma) + (vt\,\gamma + vx\,\beta\,\gamma)^2$

In[87]:= rule = $\left\{\gamma^2 \to \dfrac{1}{1-\beta^2}\right\}$;

In[88]:=
```
length5 = length5 // Simplify
```

Out[88]= $(-vt^2 + vx^2)\,(-1 + \beta^2)\,\gamma^2$

In[89]:=
```
length5 /. rule // Simplify
```

Out[89]= $vt^2 - vx^2$

Boosts: {e,p}

Length of a vector

In[90]:=
```
p1 = {e1, +p};
p1 // MatrixForm
```

Out[91]//MatrixForm=

$$\begin{pmatrix} e1 \\ p \end{pmatrix}$$

In[92]:=
```
MatrixForm /@ {p1, g, p1}
```

Out[92]=

$$\left\{\begin{pmatrix} e1 \\ p \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}, \begin{pmatrix} e1 \\ p \end{pmatrix}\right\}$$

In[93]:=
```
length = p1.g.p1
```

Out[93]= $e1^2 - p^2$

In[94]:= mRule = $\{e1^2 - p^2 \to m1^2,\ e2^2 - p^2 \to m2^2\}$

Out[94]= $\{e1^2 - p^2 \to m1^2,\ e2^2 - p^2 \to m2^2\}$

In[95]:=
```
length /. mRule
```

Out[95]= $m1^2$

In[96]:=
```
p2 = {e2, -p};
p2 // MatrixForm
```

Out[97]//MatrixForm=

$$\begin{pmatrix} e2 \\ -p \end{pmatrix}$$

In[98]:=
```
p12 = p1 + p2
```

Out[98]= {e1 + e2, 0}

Length of a rotated vector

[Note, this is a symmetric matrix, NOT anti-symmetric.]

In[99]:=
```
b2 = {{γ, β γ}, {β γ, γ}};
b2 // MatrixForm
```

Out[100]//MatrixForm=

$$\begin{pmatrix} \gamma & \beta\,\gamma \\ \beta\,\gamma & \gamma \end{pmatrix}$$

In[101]:=
```
pp2 = b2.p2;
pp2 // MatrixForm
```

Out[102]//MatrixForm=

$$\begin{pmatrix} e2\,\gamma - p\,\beta\,\gamma \\ -p\,\gamma + e2\,\beta\,\gamma \end{pmatrix}$$

In[103]:=
```
eq = pp2 == {m2, 0}
```

Out[103]= {e2 γ - p β γ, -p γ + e2 β γ} == {m2, 0}

In[104]:=
```
sol = Solve[eq // Thread, {β, γ}][[1]]
```

Out[104]= $\left\{\beta \to \frac{p}{e2},\ \gamma \to \frac{e2\,m2}{e2^2 - p^2}\right\}$

In[105]:=
```
bgRule = sol /. mRule
```

Out[105]= $\left\{\beta \to \frac{p}{e2},\ \gamma \to \frac{e2}{m2}\right\}$

In[106]:=
```
b2Lab = b2 /. bgRule;
b2Lab // MatrixForm
```

Out[107]//MatrixForm=

$$\begin{pmatrix} \frac{e2}{m2} & \frac{p}{m2} \\ \frac{p}{m2} & \frac{e2}{m2} \end{pmatrix}$$

In[108]:=
```
(b2Lab.p2 // Simplify) /. mRule
```

Out[108]= {m2, 0}

In[109]:=
```
(b2Lab.p1 // Simplify) /. mRule
```

Out[109]=

$$\left\{\frac{e1\ e2 + p^2}{m2}, \frac{(e1 + e2)\ p}{m2}\right\}$$